

Mehrere Pivot Tabellen erstellen (Teil 1)

Einleitung

- In dieser Folge programmieren wir ein Makro mit dem wir 6 Pivot Tabellen gleichzeitig erstellen (siehe Bild). Dabei wollen wir das Kundenfeedback (zu den drei Kriterien: Lieferzeit, Beratung, Gesamtzufriedenheit) auswerten.

Lieferzeit			
Anzahl von Lieferzeit			
	M	W	Gesamtergebnis
Sehr unzufrieden	1		1
Unzufrieden	5	2	7
Unentschieden	8	9	17
Zufrieden	6	17	23
Sehr zufrieden	18	11	29
Gesamtergebnis	38	39	77

Beratung			
Anzahl von Beratung			
	M	W	Gesamtergebnis
Sehr unzufrieden	2	4	6
Unzufrieden	2	4	6
Unentschieden	7	5	12
Zufrieden	7	10	17
Sehr zufrieden	20	16	36
Gesamtergebnis	38	39	77

Gesamt			
Anzahl von Gesamt			
	M	W	Gesamtergebnis
Sehr unzufrieden	3		3
Unzufrieden	4	3	7
Unentschieden	12	5	17
Zufrieden	8	12	20
Sehr zufrieden	11	19	30
Gesamtergebnis	38	39	77

Lieferzeit			
Anzahl von Lieferzeit			
	M	W	Gesamtergebnis
Sehr unzufrieden	2,6 %	0,0 %	1,3 %
Unzufrieden	13,2 %	5,1 %	9,1 %
Unentschieden	21,1 %	23,1 %	22,1 %
Zufrieden	15,8 %	43,6 %	29,9 %
Sehr zufrieden	47,4 %	28,2 %	37,7 %
Gesamtergebnis	100,0 %	100,0 %	100,0 %

Beratung			
Anzahl von Beratung			
	M	W	Gesamtergebnis
Sehr unzufrieden	5,3 %	10,3 %	7,8 %
Unzufrieden	5,3 %	10,3 %	7,8 %
Unentschieden	18,4 %	12,8 %	15,6 %
Zufrieden	18,4 %	25,6 %	22,1 %
Sehr zufrieden	52,6 %	41,0 %	46,8 %
Gesamtergebnis	100,0 %	100,0 %	100,0 %

Gesamt			
Anzahl von Gesamt			
	M	W	Gesamtergebnis
Sehr unzufrieden	7,9 %	0,0 %	3,9 %
Unzufrieden	10,5 %	7,7 %	9,1 %
Unentschieden	31,6 %	12,8 %	22,1 %
Zufrieden	21,1 %	30,8 %	26,0 %
Sehr zufrieden	28,9 %	48,7 %	39,0 %
Gesamtergebnis	100,0 %	100,0 %	100,0 %

Mehrere Pivot Tabellen erstellen (Teil 1)

Schritt 1: Mehrere Pivot Tabellen erstellen

- Im ersten Schritt erstellen wir zunächst die drei linken Pivot Tabellen (rot hervorgehoben).

```
Sub PivotTabelleErstellen()  
  
    'Variablen definieren  
    Dim pc As PivotCache 'Variable für das PivotCache  
    Dim pt As PivotTable 'Variable für die PivotTable  
    Dim ItemName As String 'Name der Tabellenüberschrift  
    Dim Zeile As Integer 'Laufvariable Tabellenblatt Analyse  
    Dim i As Integer 'Laufvariable Tabellenblatt Feedback  
  
    'Quelltabelle aktivieren  
    Worksheets("FEEDBACK").Activate  
  
    'PivotCache erstellen  
    Set pc = ActiveWorkbook.PivotCaches.Create( _  
        SourceType:=xlDatabase, _  
        SourceData:=Range("A1").CurrentRegion)  
  
    'Tabellenblatt für die PivotTabelle erstellen  
    Worksheets.Add  
    'Namen für das Tabellenblatt festlegen  
    ActiveSheet.Name = "ANALYSE"  
  
    '--- Beginn Auswertung ---  
    Zeile = 1  
  
    For i = 1 To 3 'Alle Tabellenüberschriften (Feedback) durchgehen  
  
        ItemName = Sheets("FEEDBACK").Cells(1,i+1)  
  
        '1. Überschrift im Tabellenblatt "Analyse" eintragen  
        Cells(Zeile,1).Value = ItemName  
  
        '2. Pivot Tabelle erstellen  
        Set pt = pc.CreatePivotTable( _  
            TableDestination:=ActiveSheet.Cells(Zeile+1,1))  
  
        '3.1 Felder einfügen (ZEILEN, SPALTEN)  
        pt.PivotFields(ItemName).Orientation = xlRowField  
        pt.PivotFields("Geschlecht").Orientation = xlColumnField  
        pt.DisplayFieldCaptions = False  
  
        '3.2 Felder einfügen (WERTE)  
        With pt.PivotFields(ItemName)  
            .Orientation = xlDataField  
            .Name = "Anzahl "  
            .Function = xlCount  
            .NumberFormat = "0"  
        End With  
  
        Zeile = Zeile + 10  
  
    Next i  
End Sub
```